



LibreOffice
The Document Foundation

Guia do Iniciante

Capítulo 3

Introdução às Macros

Usando o Gravador de Macros ... e mais além

Direitos Autorais

Este documento é protegido por Copyright © 2010 por seus contribuidores listados abaixo. Você pode distribuir e/ou modificar este trabalho, tanto sob os termos da Licença Pública Geral GNU (<http://www.gnu.org/licenses/gpl.html>), versão 3 ou posterior, ou da Licença de Atribuição Creative Commons (<http://creativecommons.org/licenses/by/3.0/>), versão 3.0 ou posterior.

Todas as marcas comerciais dentro desse guia pertencem aos seus respectivos donos.

Contribuidores

A versão original em inglês deste capítulo teve como autor **Ron Faile Jr.**

A tradução e localização deste capítulo teve como colaboradores:
Claudio Filho, Marcus Gama e Raul Pacheco da Silva.

Comentários e sugestões

Qualquer comentário ou sugestão sobre este documento pode ser enviado para a lista:
docs@pt-br.libreoffice.org

Agradecimentos

Este capítulo foi baseado na versão em inglês do Capítulo 13 do *Getting Started with OpenOffice.org*. Os colaboradores que contribuíram para esse capítulo foram:

Andrew Pitonyak

Jean Hollis Weber

Data de publicação e versão do software

Publicado em 11 de setembro de 2011. Baseado no LibreOffice 3.3.

Nota para usuários de Mac

Algumas teclas de atalho e itens de menu são diferentes, na plataforma Mac, daquelas utilizadas nas plataformas Windows e Linux. A tabela abaixo apresenta uma lista comparativa de alguns caminhos e teclas utilizados neste capítulo. Para uma lista completa, veja a ajuda do LibreOffice.

Windows/Linux	Equivalente no Mac	Resultado
Seleção do menu Ferramentas → Opções	LibreOffice → Preferências	Acessa as opções de configuração
<i>Clique com botão direito</i>	<i>Control+clique</i>	Abre o menu de contexto
<i>Ctrl (Control)</i>	⌘ (<i>Comando</i>)	Utilizada com outras teclas
<i>F5</i>	<i>Shift+⌘+F5</i>	Abre o Navegador
<i>F11</i>	<i>⌘+T</i>	Abre a janela <i>Estilos e formatação</i>

Sumário

<i>Direitos Autorais</i>	2
<i>Nota para usuários de Mac</i>	2
<i>Sua primeira macro</i>	4
<i>Adicionando uma macro existente</i>	4
<i>Criando uma macro simples</i>	5
<i>Executando a macro</i>	7
<i>Visualizando e editando a macro</i>	7
<i>Comentários iniciam com REM</i>	8
<i>Definindo sub-rotinas com SUB</i>	8
<i>Definindo variáveis usando DIM</i>	9
<i>Extraindo a macro em grupos</i>	9
<i>Criando uma macro</i>	10
<i>Um exemplo complicado</i>	10
<i>Executando a macro rapidamente</i>	13
<i>Às vezes o gravador de macro falha</i>	13
<i>A infraestrutura de envio (dispatch framework)</i>	13
<i>Como o gravador de macro usa a estrutura envio</i>	13
<i>Outras opções</i>	14
<i>Organização das macros</i>	14
<i>Onde as macros estão armazenadas?</i>	17
<i>Importando macros</i>	17
<i>Download de macros para importação</i>	19
<i>Como executar uma macro</i>	19
<i>Item de Menu</i>	21
<i>Teclas de atalho</i>	21
<i>Barra de ferramentas</i>	21
<i>Eventos</i>	22
<i>Extensões</i>	23
<i>Escrevendo macros sem o gravador</i>	24
<i>Encontrando mais informações</i>	24
<i>Material incluído</i>	24
<i>Recursos Online</i>	24
<i>Materiais impressos e livros eletrônicos</i>	25

Sua primeira macro

Uma macro é uma sequência de comandos salvos ou teclas digitadas que são armazenadas para uso posterior. Um exemplo de uma macro simples é aquela que “digita” seu endereço. A linguagem macro do LibreOffice é muito flexível, permitindo a automação de tarefas simples e complexas. Macros são especialmente úteis para realizar uma determinada tarefa repetidamente.

Macros do LibreOffice são geralmente escritos em uma linguagem chamada LibreOffice Basic, ou abreviadamente Basic. Embora você possa aprender e escrever macros Basic, há uma curva de aprendizagem para escrever macros a partir do zero. Os métodos usuais para um iniciante são adicionar macros que alguém já tenha escrito ou usar o gravador de macro interno, que registra suas teclas digitadas e salva-as para uso.

No LibreOffice, o gravador de macro é atualmente classificado como uma característica “experimental (instável)”. Para ativar o gravador de macro, use **Ferramentas** → **Opções** → **LibreOffice** → **Geral** e selecione a opção **Ativar recursos experimentais (instáveis)**.




Figura 1: Ativar macros na caixa de diálogo Opções

A maioria das tarefas no LibreOffice são realizadas por “despachar um comando” (enviar um comando), que é interceptado e utilizado. O gravador de macro funciona gravando os comandos que são enviados (veja “A infraestrutura de envio (dispatch framework)” na página 13).

Adicionando uma macro existente

O primeiro passo de aprendizagem na programação de macro é de encontrar e utilizar macros existentes. Esta seção assume que você tenha uma macro que você deseja usar; a macro pode estar em um e-mail, em uma página da web, ou mesmo em um livro. Por este exemplo, use a macro na Listagem 1.

Listagem 1: Simple macro que diz Olá

```
Sub Macro_01a
  Print "Olá"
End Sub
```

Você deve criar uma biblioteca e módulos para conter suas macros; que estão organizadas como descrito em “Organização das macros” na página 14. Use estas etapas para criar uma biblioteca para conter sua macro:

- 1) Use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo de Macros (ver Figura 2 e Figura 7).
- 2) Clique em **Organizador** para abrir a caixa de diálogo do Organizador de macros do LibreOffice Basic (ver Figura 8).
- 3) Selecione a aba Bibliotecas.
- 4) Definir o *Local* para *Minhas macros e caixas de diálogos*, que é o padrão.
- 5) Clique em **Nova** para abrir a caixa de diálogo Nova biblioteca. Digite na caixa *Nome* o nome para a biblioteca, como “Teste_Biblioteca” e clique **OK**.
- 6) Selecione a aba *Módulos*.
- 7) Na lista *Módulo*, expanda *Minhas macros* e selecione Teste_Biblioteca. Um módulo chamado Module1 já existe e pode conter sua macro. Você pode clicar em **Novo** para criar outro módulo em Teste_Biblioteca”.
- 8) Selecione o Module1, ou o novo módulo que você criou, e clique **Editar** para abrir Ambiente de Depuração Integrado (da sigla em inglês, IDE).
- 9) O IDE é um editor de texto para macros que permite editar e criar macros. Copie o macro no IDE.

Quando um novo módulo é criado, ele contém um comentário e uma macro chamada Main, que não faz nada.


Listagem 2: Conteúdo do Module1 após sua criação.

```
REM ***** BASIC *****  
  
Sub Main  
  
End Sub
```

Adicionar a nova macro ou antes do **Sub Main** ou após o **End Sub**. Na Listagem 3, a nova macro foi adicionada antes do **Sub Main**.

Listagem 3: Module1 após a adição da nova macro.

```
REM ***** BASIC *****  
  
Sub Macro_Ola  
  Print "Ola"  
End Sub  
  
Sub Main  
  
End Sub
```

Clique no botão Executar o BASIC na barra de ferramentas, ou pressione *F5*, para executar a primeira macro no módulo. Use a caixa de diálogo Macros do LibreOffice Basic, aberta usando o botão Selecionar macro  ou **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic**, para selecionar e executar qualquer macro no módulo.


Criando uma macro simples

Imagine a inserção repetidamente de informações simples. Embora você possa armazenar as informações na área de transferência, se você usar a área de transferência para outra coisa, o

conteúdo é alterado. Armazenar o conteúdo como uma macro é uma solução simples. (em alguns casos simples, incluindo o exemplo usado aqui, a melhor solução é a utilização de Autotexto).

- 1) Use **Ferramentas** → **Macros** → **Gravar macros** para iniciar a gravação de uma macro.

Uma pequena janela é exibida para que você saiba que o LibreOffice está gravando.



- 2) Digite as informações desejadas ou execute uma série de operações apropriadas. Neste caso, eu digitei o nome *Marcus Gama*.
- 3) Clique no botão **Para gravação** para parar a gravação, salve a macro, e a caixa de diálogo Macros do LibreOffice Basic é exibida.




Figura 2: Caixa de diálogo Macros do LibreOffice Basic no LibreOffice, módulo Writer selecionado

- 4) Esteja certo de abrir o recipiente de biblioteca chamado *Minhas macros*. Encontre a biblioteca chamada *Standard* abaixo de *Minhas macros*. Esteja ciente de que cada recipiente da biblioteca tem uma biblioteca chamada *Standard* abaixo. Selecione a biblioteca *Standard* e clique **Novo** para criar um novo módulo para conter a macro.
- 5) O nome padrão para o módulo é *Module1*. Digite um nome descritivo e clique em **OK** para criar o módulo. A caixa de diálogo Macros do LibreOffice Basic é mostrada novamente, mostrando o novo módulo.




Figura 3: Dê ao seu módulo um nome significativo

- 6) Destaque o módulo recém criado. No canto superior esquerdo (Caixa *Nome da macro*), digite o nome da macro a usar, como "DigiteMeuNome", e então clique **Novo** para salvar a macro.

Se você seguiu todos os passos, a biblioteca Padrão agora contém um módulo chamado Gravando, que contém a macro DigiteMeuNome, como mostrado na Figura 4.

Nota

Quando o LibreOffice cria um novo módulo, automaticamente adiciona a macro chamada Main.

Executando a macro

Use **Ferramentas** → **Macros** → **Executar macro** para abrir a caixa de diálogo Seletor de macro. Selecione a macro recém-criada e clique em **Executar**.




Figura 4: Selecione uma macro e clique em Executar

Existem outros métodos para executar uma macro. Por exemplo, use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic, que contém um botão **Executar** também. Outro método é usar o Organizador de macros do LibreOffice Basic; a caixa de diálogo abre mais rápida, mas o processo de seleção pode ser ligeiramente mais lento.

Visualizando e editando a macro

Você pode ver e editar a macro que acabou de ser criada. Use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic (ver Figura 4). Selecione a nova macro e clique **Editar** para abrir a macro no Basic IDE (Ambiente de Depuração Integrado).

Listagem 4: Gerando a macro "DigiteMeuNome".

```
REM ***** BASIC *****  
Sub Main
```

```

End Sub

sub DigiteMeuNome
rem -----
rem definir as variáveis
dim document as object
dim dispatcher as object
rem -----
rem ter acesso ao documento
document = ThisComponent.CurrentController.Frame
dispatcher = createUnoService("com.sun.star.frame.DispatchHelper")

rem -----
dim argsl(0) as new com.sun.star.beans.PropertyValue
argsl(0).Name = "Text"
argsl(0).Value = "Marcus Gama"

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, argsl())
end sub

```

A macro na Listagem 4 não é tão complicada como parece à primeira vista. Aprender algumas coisas ajuda significativamente na compreensão da macro gerada. A discussão começa com características próximas do topo da lista da macro descrevendo-as. Se você quiser pular os detalhes, então simplesmente mude o texto “Marcus Gama” na macro acima para o que você queira inserir na posição atual do cursor.

Comentários iniciam com REM

A palavra-chave REM, sigla de “*remark (observação)*”, começa um comentário na macro. Todo o texto depois de REM (na mesma linha) é ignorado. Como um atalho, o caractere de aspas simples também pode ser usado para iniciar um comentário.

Dica

O LibreOffice Basic não diferencia maiúsculas de minúsculas para palavra-chave, assim REM, Rem, e rem todos iniciam um comentário. Se você usa constantes simbólicas definidas pela API é mais seguro assumir que os nomes são sensíveis à caixa – constantes simbólicas são um tópico avançado e geralmente não são necessárias para pessoas que usam o gravador de macro.

Definindo sub-rotinas com SUB

Macros individuais são armazenadas em sub-rotinas definidas com a palavra-chave SUB. O final de uma sub-rotina é indicada pelas palavras END SUB. O código começa por definir a sub-rotina chamada Main, que está vazia e não faz nada. A sub-rotina seguinte, DigiteMeuNome, contém o código gerado.

Dica

O LibreOffice sempre cria uma sub-rotina vazia chamada Main quando se cria um módulo.

Há tópicos avançados que estão além do escopo deste documento, mas o conhecimento deles pode ser interessante:

- Você pode escrever uma macro de forma que os valores possam ser passados para a sub-rotina. Os valores são chamados argumentos. Macros gravadas não aceitam argumentos.

- Outro tipo de sub-rotina é chamada uma função. Uma função é uma sub-rotina que retorna um valor, A palavra-chave FUNCTION é usada em vez de SUB para definir uma função. Macros geradas são sempre do tipo SUB.

Definindo variáveis usando DIM

Você pode escrever informações em um pedaço de papel para que você possa olhá-lo mais tarde. Uma variável, como um pedaço de papel, contém informações que podem ser alteradas e lidas. A declaração DIM é semelhante a afixar à parte um pedaço de papel para ser usado para armazenar uma mensagem ou nota.

A macro DigiteMeuNome define as variáveis *document* e *dispatcher* como do tipo *object*. Outros tipos de variáveis comuns incluem *string*, *integer* e *date*. Uma terceira variável, chamado *args1*, é uma matriz de valores de propriedades. Uma variável do tipo matriz (array) permite que uma única variável contenha vários valores, semelhante ao armazenamento de várias páginas em um único livro. Valores em uma matriz são geralmente numerados a partir do zero. O número entre parênteses indica o maior número utilizável para acessar um local de armazenamento. Neste exemplo, existe apenas um valor, e é numerado como zero.

Extraindo a macro em grupos

Os detalhes a seguir são bem completos; mas não é importante compreender todos eles. A primeira linha define o início da macro.

```
sub DigiteMeuNome
```

Declara duas variáveis:

```
dim document as object
dim dispatcher as object
```

Este componente se refere ao documento atual.

A propriedade *CurrentController* de um documento se refere ao serviço que “controla” o documento. Por exemplo, quando você digita, este é o controlador atual que é acionado. O Controlador atual em seguida envia as alterações para o quadro do documento.

A propriedade *Frame* de um controlador retorna um quadro principal para um documento. Portanto, a variável chamada *document* refere-se a um quadro do documento, que recebe os comandos enviados.

```
document = ThisComponent.CurrentController.Frame
```

A maioria das tarefas no LibreOffice são realizadas enviando um comando. O LibreOffice inclui um serviço de *auxiliador de envio (dispatch helper)*, que faz a maior parte do trabalho de envio nas macros. O método *createUnoService* aceita nome de um serviço e ele tenta criar uma instância deste serviço. Concluindo, a variável *dispatcher* contém uma referência para um DispatchHelper.

```
dispatcher = createUnoService("com.sun.star.frame.DispatchHelper")
```

Declare uma matriz de propriedades. Cada propriedade tem um nome e um valor, em outras palavras, é um par nome / valor. A matriz criada tem uma propriedade no índice zero.

```
dim args1(0) as new com.sun.star.beans.PropertyValue
```

Dar a propriedade o nome “Text” e o valor “Marcus Gama”, que é o texto que é inserido quando a macro é executada.

```
args1(0).Name = "Text"
args1(0).Value = "Marcus Gama"
```

Aqui é onde a mágica acontece. O auxiliador de envio (*dispatch helper*) envia uma mensagem para o quadro do documento (armazenado na variável chamada *document*) com o comando **.uno:InsertText**. Os dois próximos argumentos, *frame name* e *search flags*, estão fora do escopo

deste documento. O último argumento é a matriz de valores de propriedades para ser usada durante a execução do comando InsertText.

```
dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args1())
```

Finalmente, o final da sub-rotina.

```
end sub
```

Criando uma macro

Ao criar uma macro, é importante fazer duas perguntas antes da gravação:

- 1) A tarefa pode ser escrita como um simples conjunto de comandos?
- 2) Etapas podem ser organizadas de tal forma que o último comando deixe o cursor pronto para o próximo comando?

Um exemplo complicado

Eu frequentemente copio linhas e colunas de dados para uma página Web e formato-os como uma tabela em um documento de texto. Primeiro, eu copio a tabela a partir da página Web para a área de transferência. Para evitar formatação e fontes estranhas, eu copio o texto em um documento do Writer como texto não formatado. Eu reformato o texto com separador entre as colunas para que eu possa usar **Tabela** → **Converter** → **De texto para tabela** para converter para uma tabela.

Eu inspeciono o texto para ver se eu posso gravar uma macro para formatar o texto (lembre-se as duas questões que eu fiz). Como exemplo, eu copiei o grupo de constantes de peso da fonte da página Web do LibreOffice. A primeira coluna indica o nome constante. Cada nome é seguido por um espaço e uma tabulação.

NÃOSEI	O peso da fonte não é especificado / conhecido.
FINO	especifica 50% do peso da fonte.
ULTRACLARO	especifica 60% do peso da fonte.
CLARO	especifica 75% do peso da fonte.
SEMICLARO	especifica 90% do peso da fonte.
NORMAL	especifica um peso de fonte normal.
SEMINEGRITO	especifica 110% do peso da fonte.
NEGRITO	especifica 150% do peso da fonte.
ULTRANEGRITO	especifica 175% do peso da fonte.
PRETO	especifica 200% do peso da fonte.

Eu quero que a primeira coluna contenha um valor numérico, a segunda coluna o nome, e a terceira coluna a descrição. O trabalho desejado é facilmente realizado para cada linha exceto para NÃOSEI e NORMAL, que não contém um valor numérico, mas eu sei que os valores são 0 e 100, por isto eu vou inserir estes valores manualmente.

O dados podem ser limpos de várias formas, em todas elas de maneira fácil. O primeiro exemplo usa atalhos de teclado que assumem que o cursor está no início da linha com o texto FINO.

- 1) Use **Ferramentas** → **Macros** → **Gravar macro** para iniciar a gravação.
- 2) Pressione **Crtl+Seta para a direita** para mover o cursor para o início da “especifica”.

- 3) Pressione *Backspace* duas vezes para remover a tabulação e o espaço.
- 4) Pressione *Tab* para adicionar a tabulação sem o espaço depois do nome constante.
- 5) Pressione *Delete* para excluir a minúscula “e” e então pressione “E” para adicionar uma letra “E” maiúscula.
- 6) Pressione *Ctrl+Seta para a direita* uma vez para mover o cursor para o início do número.
- 7) Pressione *Ctrl+Shift+Seta para a direita* para selecionar e mover o cursor antes do sinal %.
- 8) Pressione *Ctrl+C* para copiar o texto selecionado para a área de transferência.
- 9) Pressione *End* para mover o cursor para o final da linha.
- 10) Pressione *Backspace* duas vezes para remover os dois espaços à direita.
- 11) Pressione *Home* para mover o cursor para o início da linha.
- 12) Pressione *Ctrl+V* para copiar o número selecionado para o início da linha.
- 13) Colar um valor também adiciona um espaço extra, então pressione *Backspace* para remover o espaço extra.
- 14) Pressione *Tab* para inserir uma tabulação entre o número e o nome.
- 15) Pressione *Home* para mover para o início da linha.
- 16) Pressione *seta para baixo* para mover para a próxima linha.
- 17) Parar a gravação da macro e salvar a macro.

Leva muito mais tempo para ler e escrever as etapas do gravador de macro. Trabalhe lentamente e pense sobre as etapas que fará. Com a prática isto fica mais intuitivo.

A macro gerada foi modificada para conter o número de etapas nos comentários para coincidir com o código para a etapa acima.

Listagem 5: Cópia o valor numérico para o início da coluna.

```

sub CopiaNumParaColl
rem -----
rem definir as variáveis
dim document as object
dim dispatcher as object
rem -----
rem ter acesso ao documento
document = ThisComponent.CurrentController.Frame
dispatcher = createUnoService("com.sun.star.frame.DispatchHelper")

rem (2) Pressione Ctrl+Seta para a direita para mover o cursor para o início
da palavra "especifica".
dispatcher.executeDispatch(document, ".uno:GoToNextWord", "", 0, Array())

rem (3) Pressione Backspace duas vezes para remover a tabulação e o espaço.
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem -----
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem (4) Pressione Tab para adicionar a tabulação sem o espaço depois do nome
da constante.
dim args4(0) as new com.sun.star.beans.PropertyValue
args4(0).Name = "Text"
args4(0).Value = CHR$(9)

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args4())

```

```

rem (5) Pressione Delete para excluir a letra minúscula "e" ....
dispatcher.executeDispatch(document, ".uno:Delete", "", 0, Array())

rem (5) ... e então pressione "E" para adicionar uma letra "E" maiúscula.
dim args6(0) as new com.sun.star.beans.PropertyValue
args6(0).Name = "Text"
args6(0).Value = "E"

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args6())

rem (6) Pressione Ctrl+Seta para a direita uma vez para mover o cursor para o
número.
dispatcher.executeDispatch(document, ".uno:GoToNextWord", "", 0, Array())

rem (7) Pressione Ctrl+Shift+Seta para direita para selecionar o número.
dispatcher.executeDispatch(document, ".uno:WordRightSel", "", 0, Array())

rem (8) Pressione Ctrl+C para copiar o texto selecionado para a área de
transferência.
dispatcher.executeDispatch(document, ".uno:Copy", "", 0, Array())

rem (9) Pressione End para mover o cursor para o final da linha.
dispatcher.executeDispatch(document, ".uno:GoToEndOfLine", "", 0, Array())

rem (10) Pressione Backspace duas vezes para remover os dois espaços à
direita.
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem -----
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem (11) Pressione Home para mover o cursor para o início da linha.
dispatcher.executeDispatch(document, ".uno:GoToStartOfLine", "", 0, Array())

rem (12) Pressione Ctrl+V para colar o número selecionado para o início da
linha.
dispatcher.executeDispatch(document, ".uno:Paste", "", 0, Array())

rem (13) Pressione Backspace para remover o espaço extra.
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem (14) Pressione Tab para inserir uma tabulação entre o número e o nome.
dim args17(0) as new com.sun.star.beans.PropertyValue
args17(0).Name = "Text"
args17(0).Value = CHR$(9)

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args17())

rem (15) Pressione Home para mover para o início da linha.
dispatcher.executeDispatch(document, ".uno:GoToStartOfLine", "", 0, Array())

rem (16) Pressione Seta para baixo para mover para a próxima linha.
dim args19(1) as new com.sun.star.beans.PropertyValue
args19(0).Name = "Count"
args19(0).Value = 1
args19(1).Name = "Select"
args19(1).Value = false

```

```
dispatcher.executeDispatch(document, ".uno:GoDown", "", 0, args19())
end sub
```

Movimentos do cursor são usados para todas as operações (em oposição à pesquisa). Se for executado na linha NÃOSEI, a palavra *peso* é movida para a frente da linha, e o primeiro “O” é alterado para “E”. Isto não é o perfeito. Mas eu não deveria ter executado a macro sobre as linhas que não tem o formato adequado; eu preciso adaptá-las manualmente.

Executando a macro rapidamente

É tedioso executar repetidamente a macro usando **Ferramentas** → **Macros** → **Executar macro** (ver Figura 4). A macro pode ser executada a partir do IDE. Use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic. Selecione sua macro e clique **Editar** para abrir a macro no IDE.

O IDE tem um ícone **Executar o Basic** na barra de ferramentas que executa a primeira macro no IDE. A menos que você modifique a primeira macro, é a macro vazia chamado Main. Modifique a Main de modo que seja lido o que foi mostrado na Listagem 6.

Listagem 6: Modificar Main para se chamar CopiaNumParaCol1

```
Sub Main
  CopiaNumParaCol1
End Sub
```

Agora, você pode executar CopiaNumParaCol1 várias vezes clicando no ícone **Executar o Basic** na barra de ferramenta do IDE. Isto é muito rápido e fácil, especialmente para macros temporárias que serão usadas algumas vezes e depois descartadas.

Às vezes o gravador de macro falha

Compreender o funcionamento interno do LibreOffice ajuda a entender como e porquê o gravador de macro frequentemente falha. O infrator primário está relacionado com o quadro de envio e sua relação com o gravador de macro.

A infraestrutura de envio (dispatch framework)

O propósito da infraestrutura de envio (dispatch framework) é prover acesso uniforme a componentes (documentos) para comandos que normalmente correspondem a itens de menu. Eu posso usar **Arquivo** → **Salvar** do menu, a tecla de atalho *Ctrl+S*, ou clicar no ícone **Salvar** na barra de ferramentas. Todos estes comandos são traduzidos para o mesmo “comando de envio”.

O quadro de envio também pode ser usado para enviar “comandos” de volta para a IU (Interface do Usuário). Por exemplo, depois de salvar o documento, o comando Salvar Arquivo é desativado. Assim que o documento for modificado, o comando Salvar Arquivo é ativado.

Se observarmos um comando de envio, ele é um texto como *.uno:InsertObject* ou *.uno.GoToStartOfLine*. O comando é enviado para o quadro do documento, e o quadro passa o comando até que um objeto que pode manipular o comando seja encontrado.

Como o gravador de macro usa a estrutura envio

O gravador de macro geralmente grava os envios. O gravador é relativamente simples de implementar e os mesmos comandos que são emitidos são gravados para uso posterior. O problema é que nem todos os comandos de envio são completos. Por exemplo, inserir um objeto gera o seguinte código:

```
dispatcher.executeDispatch(document, ".uno:InsertObject", "", 0, Array())
```

Não é possível especificar o tipo de objeto para criar ou inserir. Se um objeto é inserido de um arquivo, você não pode especificar qual arquivo a ser inserido.

Eu gravei uma macro e usei **Ferramentas** → **Opções** para abrir e modificar itens de configuração. A macro gerada não pode registrar qualquer modificações de configuração, na verdade, o código gerado é comentado por isso não vai mesmo ser executado.

```
rem dispatcher.executeDispatch(document,  
    ".uno:OptionsTreeDialog", "", 0, Array())
```

Se uma caixa de diálogo é aberta, o comando para abrir a caixa de diálogo é susceptível de ser gerado. Qualquer trabalho feito dentro do diálogo não é gravado normalmente. Exemplos incluem diálogos de organização de macro, inserção de caracteres especiais e tipos semelhantes de diálogos. Outros possíveis problemas usando o gravador de macros incluem coisas como a inserção de uma fórmula, a definição de dados do usuário, configuração de filtros no Calc, ações em formulários de bancos de dados e exportar um documento para um arquivo criptografado PDF. Você nunca sabe ao certo o que irá funcionar a menos que você experimente-o. As ações do diálogo de pesquisa são apropriadamente capturadas, por exemplo.

Outras opções

Quando o gravador de macro não é capaz de resolver um problema específico, a solução usual é escrever o código usando os objetos LibreOffice. Infelizmente, há uma curva de aprendizagem para os objetos. Geralmente é melhor começar por exemplos simples e então diversificar lentamente à medida que você aprende mais. Aprender a ler macros geradas é um bom ponto de partida.

Se você gravar macros no Calc e o gravador pode gerar corretamente uma macro, há um suplemento criado por Paolo Mantovani, que converte macros do Calc quando elas são gravadas. O código final manipula objetos LibreOffice ao invés de gerar envios. Isto pode ser muito útil para aprender o modelo de objeto.

Você pode baixar o gravador de macros diretamente a partir da página Web do Paolo.

<http://www.paolo-mantovani.org/downloads/DispatchToApiRecorder/>

Organização das macros

No LibreOffice, as macros são agrupadas em módulos, módulos são agrupados em bibliotecas e bibliotecas são agrupadas em recipientes de bibliotecas. Uma biblioteca é normalmente usada como um agrupamento maior para uma ou outra categoria inteira de macros, ou para um aplicativo inteiro. Módulos normalmente são divididos em funcionalidades, como interação com o usuário e cálculos. Macros individuais são sub-rotinas e funções.




Figura 5: Hierarquia das bibliotecas de macro

Um cientista da computação usaria a Figura 6 para descrever precisamente a situação. O texto "1..*" significa um ou mais, e "0..*" significa zero ou mais. O losango preto significa composto de ou contém.

- Um recipiente de biblioteca contém uma ou mais bibliotecas, e cada biblioteca está contida em um recipiente de biblioteca.
- Um recipiente de biblioteca contém zero ou mais módulos, e cada módulo está contido em uma biblioteca.
- Um módulo contém zero ou mais macros, e cada macro está contida em um módulo.




Figura 6: Hierarquia da biblioteca de Macro

Use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic (ver Figura 7). Todos os recipientes de biblioteca disponíveis são mostrados na lista *Macros de*. Cada documento é um recipiente de biblioteca, capaz de conter várias bibliotecas. O aplicativo em si funciona como dois recipientes de biblioteca: um recipiente para macros distribuídas com o LibreOffice chamada Macros do LibreOffice e um recipiente para macros pessoais chamado Minhas macros. Como mostrado na Figura 7, apenas um documento está aberto no momento.




Figura 7: Recipientes da biblioteca são mostrados à esquerda

As Macros do LibreOffice são armazenadas com o código de execução do aplicativo, que podem não ser editáveis a menos que você seja um administrador. Isso é bom uma vez que estas macros não devem ser alteradas e você não deve armazenar suas próprias macros no recipiente do LibreOffice.

A menos que suas macros sejam aplicáveis a um único documento e somente para este único documento, suas macros provavelmente serão armazenadas no recipiente Minhas macros. O recipiente Minhas macros é armazenado na sua área de usuário ou diretório pessoal.

Se uma macro está contida em um documento, então uma macro gravada tentará trabalhar com aquele documento; principalmente porque ela usa “ThisComponent” para suas ações.

Cada recipiente de bibliotecas contém uma biblioteca chamada *Standard*. É melhor criar suas próprias bibliotecas com nomes significativos do que usar a biblioteca Standard. Nomes significativos não são apenas mais fáceis de gerenciar, mas eles também podem ser importados para recipientes de outra biblioteca enquanto a biblioteca Standard não pode.

Cuidado



O LibreOffice permite importar bibliotecas para um recipiente de biblioteca, mas não lhe permitirá substituir a biblioteca chamada Standard. Portanto, se você armazenar suas macros na biblioteca Standard, você não pode importá-las para um outro recipiente de biblioteca.

Assim como faz todo sentido dar a suas bibliotecas nomes significativos, é prudente usar nomes significativos para seus módulos. Por padrão, o LibreOffice usa nomes como Module1. Sinta-se livre para usar seu próprio nome significativo.

Conforme você cria suas macros, você deve decidir onde armazená-los. Armazenar uma macro em um documento é útil se o documento será compartilhado e você quiser que a macro seja incluída em seu documento. Macros armazenadas no recipiente de bibliotecas chamado Minhas macros, no entanto, são globalmente disponíveis para todos os documentos.

Macros não estão disponíveis até que a biblioteca que os contém seja carregada. As bibliotecas Standard e Template, no entanto, são carregadas automaticamente. A biblioteca carregada é exibida de maneira diferente a partir de uma biblioteca que não está carregada. Para carregar a biblioteca e os módulos que ela contém, dê um duplo clique na biblioteca.

Onde as macros estão armazenadas?

O LibreOffice armazena dados específicos do usuário em um diretório sob o diretório pessoal do usuário. A localização é específica do sistema operacional. Use **Ferramentas** → **Opções** → **LibreOffice** → **Caminhos** para ver onde outros dados de configuração estão armazenados. No Windows XP, está em `C:\Documents and Settings\\Application Data`. Macros do usuário são armazenadas em `LibreOffice\3\user\basic`. Cada biblioteca é armazenada em seu próprio diretório fora do diretório básico.

Não é importante entender onde as macros são armazenadas para uso casual. Se você sabe onde elas estão armazenadas, no entanto, você pode criar um backup, compartilhar suas macros, ou inspecioná-las se houver um erro.

Use **Ferramentas** → **Macros** → **Organizar caixas de diálogo** para abrir a caixa de diálogo Organizador de macros do LibreOffice Basic. Outra forma comum para abrir esta caixa de diálogo é a utilização de **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic e então clique no botão **Organizador**.




Figura 8: A caixa de diálogo Organizador de macros

Importando macros

A caixa de diálogo Organizador de macros do LibreOffice Basic fornece a funcionalidade para criar, excluir e renomear bibliotecas, módulos e diálogos. Selecione o recipiente de biblioteca para usar e então clique no botão **Importar** para importar uma bibliotecas de macro (ver Figura 9).

Nota

Você não pode importar a biblioteca chamada Standard.

Dica

No Linux, os arquivos específicos do LibreOffice são armazenados sob o diretório pessoal do usuário em um diretório cujo nome inicia com um ponto. Diretórios e arquivos com nomes começando com um ponto não são mostrados em uma janela de seleção normal. Para abrir o diretório, navegue para o diretório pessoal, digite o nome **.libreoffice13** e então clique **Abrir** ou pressione **Ctrl-H** para mostrar pastas e arquivos ocultos e navegar normalmente. Isto irá abrir o diretório, que não é mostrado inicialmente.




Figura 9: Selecione uma macro da biblioteca para importar

Navegue até o diretório que contém a biblioteca a ser importada. Normalmente há dois arquivos para escolher, `dialog.xlb` e `script.xlb`. Não importa qual dos dois arquivos você selecione; ambos serão importados. Selecione um arquivo e clique **Abrir** para continuar.




Figura 10: Escolha as opções de importação da biblioteca

Se a biblioteca já existe, ela não será substituída, a menos que **Substituir as bibliotecas existentes** esteja marcado. Se **Inserir como referência** estiver marcado, a biblioteca é referenciada em seu local atual, mas você não pode editar a biblioteca. Se **Inserir como referência** não estiver marcado, no entanto, a biblioteca é copiada para o diretório de macros do usuário.

Macros podem ser armazenadas em bibliotecas dentro de documentos LibreOffice. Selecione um documento ao invés de um diretório no disco (como mostrado na Figura 9) para importar bibliotecas contidas em um documento.

Download de macros para importação

Macros estão disponíveis para download. Algumas macros estão contidas em documentos, algumas como arquivos regulares que você deve selecionar e importar e algumas como texto macro que deve ser copiado e colado no IDE Basic; use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic, escolha a macro para editar, e então clique **Editar** para abrir a macro no IDE Basic.

Algumas macros estão disponíveis como download gratuitos na Internet (ver Tabela 1). “Adicionando uma macro existente” na página 4 descreve como adicionar essas macros à sua biblioteca de macros

Tabela 1. Lugares para encontrar exemplos de macros

Localização	Descrição
http://www.oocomacros.org/	Coleção de macros empacotadas – um pouco antigas agora.
http://www.pitonyak.org/oo.php	Materiais de referência sobre macros.
http://www.pitonyak.org/database/	Materiais de referência sobre macros de banco de dados.
http://development.openoffice.org/	Coleção de links para tudo.
http://www.oooforum.org/	Um fórum, com muitos exemplos e ajuda.
http://user.services.openoffice.org/	Outro fórum, com muitos exemplos e ajuda.

Como executar uma macro

Um método típico para executar uma macro é a seguinte:

- 1) Use **Ferramentas** → **Macros** → **Executar macro** para abrir a caixa de diálogo Seletor de macro (ver Figura 11).
- 2) Selecione a biblioteca e o módulo na lista Biblioteca (do lado esquerdo)
- 3) Selecione a macro na lista Nome da macro (do lado direito)
- 4) Clique **Executar** para executar a macro.




Figura 11: Usar a caixa de diálogo Seleção de macro para executar macros

Embora você possa usar **Ferramentas** → **Macros** → **Executar macro** para executar todas as macros, isto não é eficiente para macros frequentemente executadas. A técnica mais comum é atribuir a macro a um botão na barra de ferramentas, item de menu, tecla de atalho, ou um botão incorporado em um documento. Ao escolher um método, também é bom fazer perguntas tais como:


- A macro está disponível apenas para um documento ou globalmente para todos os documentos?
- A macro pertence a um tipo de documento específico, como um documento do Calc?
- Com que frequência a macro vai ser usada?

As respostas vão determinar onde armazenar a macro e como torná-la disponível. Por exemplo, você provavelmente não irá adicionar uma macro raramente utilizada a uma barra de ferramentas. Para ajudar a determinar suas escolhas, veja a Tabela 2.

Tabela 2. Métodos para iniciar uma macro

Tipo	LibreOffice	Tipo de documento	Documento
Barra de ferramentas	Não	Sim	Sim
Menu	Não	Sim	Sim
Atalho	Sim	Sim	Não
Evento	Sim	Não	Sim

Para adicionar um item de menu, tecla de atalho, ou ícone na barra de ferramentas que chame uma macro, use a caixa de diálogo Personalizar (ver Figura 13). Abra esta caixa de diálogo usando qualquer um dos seguintes métodos:

- Escolha **Ferramentas** → **Personalizar** da barra de menu principal.
- Cada barra de ferramentas tem um ícone  que abre um menu; escolha a opção **Personalizar barra de ferramentas**.

Dica

A cobertura completa da caixa de diálogo Personalizar está além do escopo deste documento. Clique no botão **Ajuda** para acessar a página de ajuda que acompanha o LibreOffice.

A caixa de diálogo Personalizar contém abas para configurar menus, atalhos de teclado, barra de ferramentas, e eventos.

Item de Menu

Use **Ferramentas** → **Personalizar** para abrir a caixa de diálogo Personalizar, e selecione a aba Menus. Você pode modificar um menu existente ou criar novos menus que chamam macros. Para mais informações sobre menus modificados, consulte o Capítulo 14.

Teclas de atalho

Use **Ferramentas** → **Personalizar** para abrir a caixa de diálogo Personalizar, e selecione a aba Teclado. A atribuição de atalhos de teclado é discutida no Capítulo 14.




Figura 12: Caixa de diálogo Personalizar do LibreOffice

Barra de ferramentas

Macros podem ser adicionadas às barras de ferramentas. Para mais informações sobre como modificar barras de ferramentas, consulte o Capítulo 14, Personalizando o LibreOffice.

Eventos

No LibreOffice, quando algo acontece, dizemos que ocorreu um evento. Por exemplo, um documento foi aberto, uma tecla foi pressionada ou o mouse foi movido. O LibreOffice permite que eventos causem um chamado de macro; a macro é então chamada por um processador de eventos. A cobertura total sobre o processador de eventos está além do escopo deste documento, mas um pouco de conhecimento pode ajudar muito.

Cuidado



Tenha cuidado quando você configura um processador de eventos. Por exemplo, suponha que você escreve um processador de evento que é chamado a cada vez que uma tecla é pressionada, mas você cometeu um erro, de modo que o evento não é tratado corretamente. Um resultado possível é que seu processador de evento consumirá todas as teclas pressionadas, forçando você a finalizar à força o LibreOffice.

Use **Ferramentas** → **Personalizar** para abrir a caixa de diálogo Personalizar, e selecione a aba Eventos (ver Figura 13). Os eventos na caixa de diálogo Personalizar estão relacionados com a aplicação completa e documentos específicos. Use a caixa Salvar em para escolher LibreOffice, ou um documento específico.




Figura 13: Atribuir macro a um evento de nível de aplicativo

Um uso comum é atribuir o evento Ao criar um novo documento para chamar uma macro específica. A macro então executa certas tarefas de configuração para o documento. Selecione o evento desejado e clique no botão **Macro** para abrir a caixa de diálogo Seletor de macro (ver Figura 14).




Figura 14: Atribuir macro ao evento documento aberto

Selecione a macro desejada e clique **OK** para atribuir a macro ao evento. A aba Eventos mostra que o evento foi atribuído para a macro (ver Figura 15). Quando um documento é criado, a macro DigiteMeuNome é executada.

Muitos objetos em um documento podem ser definidos para chamar macros quando ocorrem eventos. O uso mais comum é adicionar um controle, como um botão, em um documento. Mesmo um duplo clique sobre um gráfico abre a caixa de diálogo com uma aba de Macros, que permite atribuir uma macro a um evento.




Figura 15: Macro DigiteMeuNome é atribuída ao evento Ao criar um novo documento

Extensões

Uma extensão é um pacote que pode ser instalado no LibreOffice para adicionar novas funcionalidades. Extensões podem ser escritas em praticamente qualquer linguagem de programação e podem ser simples ou sofisticadas. Extensões podem ser agrupadas em tipos:

- Suplementos Calc, que oferecem novas funcionalidades para o Calc, incluindo novas funções que agem como funções internas normais.
- Novos componentes e funcionalidades, que normalmente incluem algum nível de integração com a interface do usuário, tais como novos menus e barras de ferramentas.
- Dados pilotos que são usados diretamente no Calc.
- Suplementos gráficos com novos tipos de gráficos.
- Componentes de linguística, como um corretor ortográfico.
- Modelos de documentos e imagens.

Embora extensões individuais podem ser encontradas em diferentes lugares, existe atualmente um repositório de extensões em <http://extensions2.libreoffice.org/> e alguma documentação em <http://libreplanet.org/wiki/Group:OpenOfficeExtensions/List>.

Para mais informações sobre como obter e instalar extensões, consulte o Capítulo 14, Personalizando o LibreOffice.

Escrevendo macros sem o gravador

Os exemplos abordados neste capítulo são criados usando o gravador de macros e o *dispatcher*. Você também pode escrever macros que acessem diretamente os objetos que compõem o LibreOffice. Em outras palavras, você pode manipular diretamente um documento.

Manipular diretamente os objetos internos do LibreOffice é um tópico avançado que está além do escopo deste capítulo. Um exemplo simples, no entanto, demonstra como isso funciona.

Listagem 7: Acrescentar o texto “Olá” para o documento atual.

```
Sub AdicionaOla
    Dim oDoc
    Dim sTextService$
    Dim oCurs

    REM Este componente refere-se ao documento ativo no momento.
    oDoc = ThisComponent

    REM Verifica se este é um documento de texto.
    sTextService = "com.sun.star.text.TextDocument"
    If NOT oDoc.supportsService(sTextService) Then
        MsgBox "Esta macro somente funciona com documentos de texto"
        Exit Sub
    End If

    REM Obter o cursor visto a partir do controle atual.
    oCurs = oDoc.currentController.getViewCursor()

    REM Move o cursor para o final do documento.
    oCurs.gotoEnd(False)

    REM Insere o texto "Ola" no final do documento.
    oCurs.Text.insertString(oCurs, "Olá", False)
End Sub
```

Encontrando mais informações

Inúmeros recursos estão disponíveis que fornecem ajuda com macros escritas. Use **Ajuda** → **Ajuda do LibreOffice** para abrir a página de ajuda do LibreOffice. O canto superior esquerdo do sistema de ajuda do LibreOffice contém uma lista suspensa que determina que texto de ajuda é mostrado. Para exibir a ajuda para o Basic, escolha *LibreOffice Basic* nesta lista.

Material incluído

Muitas excelentes macros são incluídas com o LibreOffice. Use **Ferramentas** → **Macros** → **Organizar macros** → **LibreOffice Basic** para abrir a caixa de diálogo Macros do LibreOffice Basic. Expanda a biblioteca *Tools* no recipiente de biblioteca LibreOffice. Explore o módulo *Debug* – alguns bons exemplos incluem o `WritedbInfo(documento)` e o `printdbgInfo(planilha)`.

Recursos Online

Os links e referências a seguir contém informações sobre a programação de macros:

<http://user.services.openoffice.org/> (fóruns do Ooo, com bom suporte; os voluntários respondem questões sobre o LibreOffice também).

<http://api.openoffice.org/docs/common/ref/com/sun/star/module-ix.html> (IDL oficial de referência; aqui você encontrará praticamente todos os comandos com uma descrição).

http://wiki.services.openoffice.org/wiki/Documentation/BASIC_Guide (Guia de Programação Basic Oficial do OpenOffice.org) Também disponível em formato ODT em http://wiki.documentfoundation.org/Documentation#Other_Documentation_and_Resources

<http://wiki.services.openoffice.org/wiki/Documentation/DevGuide> (Guia de Desenvolvimento Oficial do OpenOffice.org; contém uma explicação detalhada). Também disponível em formato ODT em http://wiki.documentfoundation.org/Documentation#Other_Documentation_and_Resources

<http://www.pitonyak.org/oo.php> (Página Andrew Pitonyak de macros)

<http://www.pitonyak.org/AndrewMacro.odt> (numerosos exemplos de macros de trabalho).

http://www.pitonyak.org/OOME_3_0.odt (Livro de Andrew Pitonyak sobre macros)

<http://www.pitonyak.org/database/> (numerosos exemplos de macro usando o Base).

<http://docs.sun.com/app/docs/doc/819-0439> (Livro da Sun sobre programação de macro, muito bem escrito e apresentado; O Guia de Programação BASIC do Ooo e o Guia de Desenvolvimento do Ooo são derivados deste livro).

Materiais impressos e livros eletrônicos

Atualmente não há livros específicos para macros do LibreOffice que estão disponíveis para download.

Para o OpenOffice.org, os seguintes livros estão disponíveis para compra na forma impressa e eletrônica de seus editores.

Dr. Mark Alexander Bain's *Learn OpenOffice.org Spreadsheet Macro Programming*.

Ver <http://www.packtpub.com/openoffice-ooobasic-calc-automation/book>.

Roberto Benitez's *Database Programming with OpenOffice.org Base & Basic*.

Ver <http://www.lulu.com/product/paperback/database-programming-with-openofficeorg-base-basic/3568728>